# This file is part of BlackArch Linux ( https://www.blackarch.org/ ).
# See COPYING for license details.

pkgname=arybo
pkgver=65.89d9a42
pkgrel=9
pkgdesc='Manipulation, canonicalization and identification of mixed boolean-arithmetic symbolic expressions.'
groups=('blackarch' 'blackarch-misc')
arch=('any')
url='https://github.com/quarkslab/arybo'
license=('BSD')
depends=('python' 'python-networkx' 'python-six' 'python-pytanque' 'ipython')
makedepends=('git' 'python-setuptools')
source=("git+https://github.com/quarkslab/$pkgname.git")
sha512sums=('SKIP')

pkgver() {
  cd $pkgname

  echo $(git rev-list --count HEAD).$(git rev-parse --short HEAD)
}

build() {
  cd $pkgname

  python setup.py build
}

package() {
  cd $pkgname

  python setup.py install --root="$pkgdir" --prefix=/usr -O1 --skip-build

  install -Dm 644 -t "$pkgdir/usr/share/doc/$pkgname/" README.rst CHANGELOG \
    CONTRIBUTORS
  install -Dm 644 LICENSE.txt "$pkgdir/usr/share/licenses/$pkgname/LICENSE"
}

